CISPA
HELMHOLTZ CENTER FOR
INFORMATION SECURITY

Formal Analysis of Real-World Security Protocols

Lecture 3: Attacker Model and Trace Properties

Aleksi Peltonen | 11.11.2024

Exercises

- We have finished grading the first exercise sheet
- You will receive feedback through CMS later today
- Questions about the grading? Send an email to `alexander.dax@cispa.de`
  - Include your **team number** and question
  - e.g., "I am member of Team #1. I do not understand the feedback we got for Exercise sheet 1, Ex. 1b. Could you please clarify what our mistake was?"
- The next exercise sheet will be published today at 16:00
  - Feel free to ask questions on the **forum** if something is unclear!

Model components

What **components** do we need to model protocols?

1.	All possible sent and received messages	Lecture 1
2.	All possible protocol behaviors	Lecture 2
3.	The attacker	Lecture 3
4.	Security properties that we want to verify	

This lecture

Actions and Action Traces

Protocol Model

Attacker Model

Trace Properties

Actions and Action Traces

Action facts

- **Actions**, like regular facts, are built from predicates applied to terms
- They model actions taken by agents during protocol execution and steps taken during protocol initialization

```
// Send message
[ Fr(~m) ] --[ Send(~m) ]-> [ Out(~m) ]

// Receive message
[ In(m)  ] --[ Receive(m) ]-> [ ]
```

- Actions are analogous to labels in labelled transition systems and can be used for **property specification**

Executions

An execution is a sequence of steps that a system can take, starting from an initial state and following a set of rules. Each step transforms the current state into a new state by applying a specific rule.

- Let R be a set of rules constructed over a given signature, and let S be a state of the system, i.e., a multiset of facts
- An **execution** of R with respect to an equational theory E is an alternating sequence of states and ground rule instances:

$$[\, S_0, l_1 -[\ a_1\ ]\!\rightarrow r_1, S_1, l_2 -[\ a_2\ ]\!\rightarrow r_2, \ldots, S_{k-1}, l_k -[\ a_k\ ]\!\rightarrow r_k, S_k \,]$$

such that the following three conditions hold:

1. $S_0 = [\,]$,
2. $\forall i \in \{1 \ldots k\}, (S_{i-1}, (l_i -[\ a_i\ ]\!\rightarrow r_i), S_i) \in steps(R)$, and
3. $\forall i, j \in \{1 \ldots k\}, r_i = [\,] -[\ \ ]\!\rightarrow [Fr(n)]$ and $r_j = [\,] -[\ \ ]\!\rightarrow [Fr(n)]$: $i = j$.

- We denote the set of executions of a set of rules R by $execs(R)$

Traces

- For each execution, we define the corresponding trace as the sequence $[set(a_1), set(a_2), \ldots, set(a_k)]$ and denote the set of all traces of a set of rules R by $traces(R)$
- Consider the following protocol:

```
[      ] --[ Init()  ]-> [ A('1') ] // Create A('1')
[ A(x) ] --[ Step(x) ]-> [ B(x)   ] // Convert A(x) to B(x)
```

- One possible execution:

```
[                ] --[ Init()    ]->
[ A('1')         ] --[ Init()    ]->
[ A('1'), A('1') ] --[ Step('1') ]->
[ A('1'), B('1') ]
```

- Corresponding trace:

```
[ Init(), Init(), Step('1') ]
```

Example

Condition 1: Initial State
What it says: The system starts from an empty state.
What it means: At the very beginning (time S), there are no facts in the system. It's like starting with a blank slate.

Condition 2: Valid Transitions
What it says: Each step in the sequence must:

Start from the current state (S).
Apply a rule (a) from the set of rules (R).
End in a new valid state (S).
What it means:

The system can only move forward if a rule allows it.
Every state transition must make sense according to the rules in R.

Condition 3: each rule application is unique and doesn't produce conflicting outputs:
If i=j it's the same rule, so it should produce consistent results.
If i != jthey are different rules, and their outputs should be independent.

Needham-Schroeder Public-Key protocol (NSPK)



Protocol model



Initialization


```
builtins: asymmetric-encryption


/* Public key infrastructure */
rule register_pk:
    let
      public_key = pk(~secret_key)
    in
    [ Fr(~secret_key) ]
    -->
    [ !Sk($ID, ~secret_key)
    , !Pk($ID, public_key)
    , Out(public_key) ]


/* Reveal secret key */
rule reveal_sk:
    [ !Sk(ID, secret_key) ]
  --[ Reveal(ID) ]->
    [ Out(secret_key) ]
```

Initiator (1/2)




```
/* Generate a fresh nonce nI and
   send an encrypted message
   to R. */
rule initiator_1:
    let
      m1 = aenc{'1', ~nI, $I}pkR
    in
    [ Fr(~nI)
    , !Pk($R, pkR) ]
  --[ Send(m1) ]->
    [ Out(m1)
    , St_I_1($I, $R, ~nI) ]
```

Responder (1/2)




```
/* Receive an encrypted message
   from I and decrypt it.
   Derive a fresh nonce nR and
   reply to I. */
rule responder_1:
    let
      m1 = aenc{'1', nI, I}pk(skR)
      m2 = aenc{'2', nI, ~nR}pkI
    in
    [ In(m1)
    , !Sk(R, skR)
    , !Pk(I, pkI)
    , Fr(~nR) ]
  --[ Receive(nI, m1)
    , Send(m2)
    , Running(R, I, ~nR, nI) ]->
    [ Out(m2)
    , St_R_1(R, I, nI, ~nR) ]
```

Initiator (2/2)




```
/* Receive an encrypted message
   from R and decrypt it.
   Respond to R. */
rule initiator_2:
    let
      m2 = aenc{'2', nI, nR}pk(skI)
      m3 = aenc{'3', nR}pkR
    in
    [ In(m2)
    , St_I_1(I, R, nI)
    , !Sk(I, skI)
    , !Pk(R, pkR) ]
  --[ Receive(nR, m2)
    , Running(I, R, nI, nR)
    , Commit(I, R, nI, nR) ]->
    [ Out(m3) ]
```

Responder (2/2)




```
/* Receive a message from I. */
rule responder_2:
    let
      m3 = aenc{'3', nR}pk(skR)
    in
    [ In(m3)
    , St_R_1(R, I, nI, nR)
    , !Sk(R, skR) ]
  --[ Commit(R, I, nR, nI) ]->
    [  ]
```

Protocol Model

Protocol model in Tamarin

- **Term algebra**
  - $\Sigma_{DH} = \{enc(\_,\_), dec(\_,\_), h(\_), \langle\_,\_\rangle, fst(\_), snd(\_), \_\hat{}\_, \_^{-1}, \_ \times \_, 1\}$
- **Equational theory**
  - $E_{DH} = \{dec(enc(m,k),k) =_E m, x \times (y \times z) =_E (x \times y) \times z, \ldots\}$
- **Facts**
  - $F(t_1, \ldots, t_n)$
- **Transition system**
  - State: multiset of facts
  - Rules: $l \dashv\![\ a\ ]\!\mapsto r$
- **Special facts and rules**
  - Facts: $In()$, $Out()$, $K()$
  - Special fresh rule: $[\,] \dashv\![\quad]\!\mapsto [\,Fr(x)\,]$

Semantics

- **Transition relation**

  $S \dashv\![\ a\ ]\!\!\rightarrowtail_R ((S \setminus^\# l) \cup^\# r)$, where

  - $l \dashv\![\ a\ ]\!\!\rightarrowtail r$ is a ground instance of a rule in R, and
  - $l \subseteq^\# S$ wrt the equational theory

- **Executions**

  - $execs(R) = \{\, [\,] \dashv\![\ a_1\ ]\!\!\rightarrowtail \ldots \dashv\![\ a_n\ ]\!\!\rightarrowtail S_n \mid \forall n.\ Fr(n)$ appears only once on the right-hand side of the rule $\}$

- **Traces**

  - $traces(R) = \{\, [a_1, \ldots, a_n] \mid [\,] \dashv\![\ a_1\ ]\!\!\rightarrowtail \ldots \dashv\![\ a_n\ ]\!\!\rightarrowtail S_n \in execs(R)\,\}$

Attacker Model

Attacker model

Recall the **protocol execution model** from earlier:



The Dolev-Yao model

- All messages are sent to the attacker who can either **drop**, **modify**, or **forward** them
- The attacker sees all the messages and maintains a knowledge set of all the information sent over public channels
- When the attacker learns a cryptographic key, it can perform cryptographic operations, such as **encryption**, **decryption**, and **signing**, to add new messages to its knowledge set
- The attacker can also deconstruct messages into their components and create new messages from the parts it knows
- However, it cannot forge or read cryptographically protected messages **without knowing the corresponding keys**

Some potential attacks

Man-in-the-middle: *c* impersonates *a* to *b*

Replay: reuse previous messages

Reflection: send message back to its sender

Oracle: use normal protocol responses to gain information

Binding: use messages in an unintended context

Type flaw: substitute message fields

Attacker knowledge and interaction

- A persistent fact **K(m)** denotes that *m* is known to the adversary
- A linear fact **Out(m)** denotes that the protocol has sent the message *m*, which can be received by the adversary
- A linear fact **In(m)** denotes that the protocol can receive the message *m*, which might have been sent by the attacker
- The semantics of these three fact symbols is given by the following set of **message deduction rules**

Message deduction rules

$$\left\{\frac{\mathrm{Out}(x)}{\mathrm{K}(x)}\right\}$$

```
// Receive message from the protocol
[ Out(x) ] --> [ K(x) ]
```

$$\left\{\frac{\mathrm{K}(x)}{\mathrm{In}(x)}[\mathrm{K}(x)]\right\}$$

```
// Send message to the protocol
[ K(x) ] --[ K(x) ]-> [ In(x) ]
```

$$\left\{\frac{}{\mathrm{K}(x : \mathrm{pub})}\right\}$$

```
// Learn public value
[ ] --> [ K($x) ]
```

$$\left\{\frac{\mathrm{Fr}(x : \mathrm{fresh})}{\mathrm{K}(x : \mathrm{fresh})}\right\}$$

```
// Generate fresh value
[ Fr(~x) ] --> [ K(~x) ]
```

$$\left\{\frac{\mathrm{K}(x_1)\ldots\mathrm{K}(x_k)}{\mathrm{K}(f(x_1\ldots x_k))}\right\}$$

```
// Apply functions to known messages
[ K(x_1)...K(x_k) ] --> [ K(f(x_1...x_k)) ]
```

Trace Properties

Trace properties

- A trace property specifies a set of traces representing a set of desired protocol behaviors
- If the protocol state machine includes behaviors that are not included in the specified property, then we have a violation
  → **This constitutes an attack on the protocol!**
- In Tamarin, trace properties are specified as formulas in **first-order logic**, built from actions and quantifying over message terms and timepoints
- Timepoints are are used to order actions; they enable the specification of properties that depend on the events' relative ordering

Syntax

`All`	Universal quantification ($\forall$)
`Ex`	Existential quantification ($\exists$)
`==>`	Implication
`&`	Conjunction
`|`	Disjunction
`not`	Negation ($\neg$)
`f@i`	An action `f` at a timepoint `#i`
`#i < #j`	Timepoint `#i` occuring before `#j`
`#i = #j`	Timepoint equality
`x = y`	Message variable equality
`Pred`$(t_1, \ldots, t_n)$	The predicate `Pred` applied to the terms t_1 to t_n

Example

Needham-Schroeder Public-Key protocol (NSPK)



Lemma 1: Executability

To rule out (some) modeling mistakes, we use reachability lemmas to make sure that it is **possible** to reach the end of the protocol model. Our goal is to find a completed protocol trace where the steps are the expected ones taken by honest agents without adversary interference.

```
/* Executability */
lemma trace: exists-trace
    " Ex A B nA nB #i #j .
        Commit(A, B, nA, nB)@i
      & Commit(B, A, nB, nA)@j
      & not(Ex #r. Reveal(A)@r)
      & not(Ex #r. Reveal(B)@r)
    "
```

Lemma 2: Injective agreement

Whenever somebody commits to running a session and the adversary did not reveal the long-term key of the participants, there is somebody running a session with the same parameters and there is no other commit on the same parameters.

```
/* Injective agreement */
lemma injective_agreement:
    " All A B nA nB #i .
        Commit(A, B, nA, nB)@i
        ==> (Ex #j. Running(B, A, nB, nA)@j & j < i
                & not(Ex A2 B2 #i2 .
                    Commit(A2, B2, nA, nB)@i2 & not(#i = #i2)))
            | (Ex #r. Reveal(A)@r)
            | (Ex #r. Reveal(B)@r)
    "
```

Summary

Next lecture

- We now know how to model..
  - ..protocol behavior as **multiset rewriting rules**
  - ..protocol properties as **first-order logic formulas**
- Together, these two languages allow us to **model protocols**, **specify security properties**, and **analyze them in the presence of a Dolev-Yao attacker**
- In the next lecture, we will talk about how Tamarin uses this model to find attacks

Reading material

Recommended reading:
[Bas+24, Ch. 4.2.2, 5, 6–6.4], [Mei13, Ch. 7.3], [Sch+12]

[Bas+24] D. Basin, C. Cremers, J. Dreier, and R. Sasse. **Modeling and Analyzing Security Protocols with Tamarin: A Comprehensive Guide.** Draft v0.5. Sept. 2024.

[Mei13] S. Meier. **Advancing Automated Security Protocol Verification.** PhD thesis. ETH Zurich, 2013.

[Sch+12] B. Schmidt, S. Meier, C. Cremers, and D. Basin. **Automated Analysis of Diffie-Hellman Protocols and Advanced Security Properties.** In: 2012 IEEE 25th Computer Security Foundations Symposium. 2012.